

**ЎЗБЕКИСТОН РЕСПУБЛИКАСИ ОЛИЙ ВА ЎРТА МАХСУС
ТАЪЛИМ ВАЗИРЛИГИ**

НАМАНГАН ДАВЛАТ УНИВЕРСИТЕТИ

КОМПЮТЕРДА ФИЗИКА ТАЖРИБАЛАРИ

(УСЛУБИЙ КЎРСАТМА)

П.Ж. Байматов, Н.Рахимов, А. Пўлатов,
О.Ахмедов, Ш.Иноятов

**ИККИ ЎЛЧОВЛИ ЯШИКДАГИ
ЗАРЯДЛАНГАН ЗАРРАЧАЛАР
ХАРАКАТИНИНГ СОДДА
МОДЕЛИ**

НАМАНГАН 2007

**УШБУ ЎҚУВ УСЛУБИЙ КЎРСАТМА НАМДУ ЎҚУВ УСЛУБИЙ
КЕНГАШИНИНГ “ ” _____ 2008 ЙИЛДАГИ ЙИҒИЛИШИДА
ТАСДИҚЛАНГАН**

БАЁННОМА № ____

**ЎҚУВ УСЛУБИЙ КЎРСАТМА НАМДУ ФИЗИКА МАТЕМАТИКА
ФАКУЛЬТЕТИ УСЛУБИЙ КЕНГАШИНИНГ “ 29 ” ФЕВРАЛ 2008 ЙИЛДАГИ
ЙИҒИЛИШИДА ТАСДИҚЛАНГАН**

БАЁННОМА № 7

**ЎҚУВ УСЛУБИЙ КЎРСАТМА ФИЗИКА КАФЕДРАСИ УСЛУБИЙ
КЕНГАШИНИНГ “28” ЯНВАР 2008 ЙИЛДАГИ ЙИҒИЛИШИДА
ТАСДИҚЛАНГАН**

БАЁННОМА № 6

Такризчилар :

А. Шералиев наманган мухандислик
Педагогика институти доценти

Х. Қўчқаров НАМДУ физика
кафедраси доценти

СЎЗ БОШИ

Ушбу ўқув услубий кўрсатма Термодинамика ва Статистик физиканинг мухим масалаларидан бўлиб бакалавр даражасини олиш учун тахсил олаётган талабалар учун мўлжалланган. Хозирги замон талаб ва таклифларидан келиб чиқиб кадрлар тайёрлаш миллий дастурида кўрсатилгандек замонавий таълим технологияларидан фойдаланган холда эски усулда дарс ўтишдан воз кечилиб янги ноанъанавий холда дарс ўтиш методларини ишлаб чиқишга қарор қилдик. Бунинг учун икки ўлчовли яшиқдаги зарядланган заррани компьютер моделини яратдик. Бу моделда талаба заррачани харакатини оддий сўз билан эшитиб эмас балки унинг реал кўринишадги харакатини гувоҳи бўлади. Бундай тажриба талабани фикрлашига ҳамда дарсни тушунига анча яхши самара беради. Ушбу услубий қўлланмадан нафақат бакалавр талабаси балки магистлар ҳам фойдаланса мақсадга мувофиқ бўлар эди.

МУАЛЛИФЛАР

1. Кириш. Икки ўлчовли ҳаракат ҳақида
2. Бошланғич шартлар: координата ва тезликлар
3. Ҳаракат тенгламасини интеграллаш
4. Заррачаларни яшиқ деворидан кайтиши
5. Экранда тасвирлаш. Мултипликация

Саволлар

Адабиётлар

Илова. Моделнинг Паскал тилидаги тула дастури

1. Кириш. Икки ўлчовли ҳаракат ҳақида

Табиатда газ, суюқлик, қаттиқ жисм каби моддалар кўп микроразрачалардан иборат бўлиб бу заррачалар мураккаб қонунлар бўйича узаро таъсирлашувда бўлади. Агар моддалардаги заррачалар сонини тасаввур қилсак ($\sim 10^{20} \dots 10^{22} \text{ см}^{-3}$) ҳар бир зарра ҳаракатини кузатиш учун 10^{20} тартибдаги Нютон тенгламалар системасини ечишга тугри келади. Бунининсон қулда бажариш мумкин эмас. Ҳозирги кунга келиб компютер ресурслари $\sim 10^5$ ва ундан юқори тартибдаги тенгламалар системасини интеграллашга имкон беради.

Заррачалараро таъсирлашув қонунларини билган ҳолда газлардаги молекулалар ҳаракатини компютерда моделлаштириш мумкин. Бунда заррачалар ҳаракати классик механика қонунларига бойсунади деб қаралади. Зарралараро таъсирлашувни молекуляр физикада маълум бўлган феноменологик формулалар орқали ифодаланади. Масалан Леонард-Жонс потенциали

$$V(r) = 4\epsilon [(\sigma/r)^{12} - (\sigma/r)^6] \quad (1)$$

кўпгина реал газларнинг молекулаларини таъсирлашув табиатига мосдир.


Соддалаштирилган системаларни ўрганиш - моделлаш методларини ўрганишда дастлабки мухим кадам ҳисобланади. Бунга мисол килиб икки ўлчовли яшикка камалган зарядли заррачаларни олиш мумкин. Зарядлар узаро Кулон

$$V(r) = q_1 q_2 / r = q^2 / r = Z^2 / r \quad (2)$$

($Z^2 = q_1 q_2 / r = q^2 / r$) қонунига асосан итаришиши туфайли узаро узоклашиши табиий. Лекин яшик деворлари эластик қайтарувчи деб ҳисобласак (қаттиқ чегаравий шарт) жуда содда модел ҳосил бўлади. Албатта, яшик деворлари билан тўқнашганда бошқача (даврий чегаравий) шартлардан фойдаланилса модел янада мукаммаллашади.

Зарраларни икки ўлчовли ҳаракатига кўплаб реал мисоллар келтириш мумкин. Масалан қаттиқ жисмларда сиртий ўтказувчанлик, Ута анизотроп кристалларди атомлар тебраниши ёки электронлар ҳаракати, гетеропереход лардаги квантланган электрон газлар в.х. Лекин бундан уч ўлчовли системаларни моделини яратиш мураккаб экан деб ҳисобламаслик керак. Бизни асосий мақсадимиз: икки ўлчовли содда система орқали компьютерда моделлаш методлари билан ўқувчини

таништиришдир. Бунда ўқувчи зарралар харакатини дисплей экраннда бевосита визуал кузатишни дастурлашда кийналмайди.



Биз ўқувчини: Нютонни харакат тенгламаси, биринчи тартибли дифференциал тенгламани такрибий интеграллаш (Эйлер методи) ва Паскал тили билан кисман таниш деб хисоблаймиз. Лекин бу катъий талаб эмас. Куйида бу тушунчаларни мисоллар жараёнида имкон борича ойдинлаштириб боришга харакат килинди.

2. Бошланғич шартлар: координата ва тезликлар

Дисплей экранни икки ўлчовли яшик деб оламиз, у холда яшик кенглигини `GetMaxX` - экранни максимал кенглигини кийматини берувчи оператор, яшик баландлигини эса `GetMaxY` - экранни максимал баландлигини кийматини берувчи оператор

ёрдамида ифодалаш мумкин. Дисплей координаталари йуналиши расмдаги куринишда булади.

Куйидаги дастур координата ва тезликларнинг Бошланғич кийматларини киритади

(D1)

For i:=1 to N do

begin x[i]:= 10 + Random(GetMaxX - 10);

y[i]:= 10 + Random(GetMaxY - 10);

Vx[i]:=Vm*(2*0.01*Random(100) - 1);

Vy[i]:=Vm*(2*0.01*Random(100) - 1);

Xold [i]:=x[i];

Yold[i]:= y[i];

end;

Яшиқдаги зарралар сони N та десак у холда зарраларни координаталари ва тезликларининг x,y проекциялари массивини x[i], y[i], Vx[i], Vy[i] каби белгиланган. Максимал тезлик сифатида Vm киритилган. Хисоблаш давомида ёрдамчи узгарувчи сифатида зарраларнинг эски координаталарини саклаб колувчи массив Xold[i], Yold[i] хам киритилган.

Random(100) оператори 0..99 оралигида тасодикий сонни беради, у холда

$$0.01 * \text{Random}(100)$$

ифода 0..0,99 оралигида кийматни

$$2 * 0.01 * \text{Random}(100) - 1$$

ифода эса тахминан -1..+1 оралигида кийматни беради. Демак

$V_x[i]$, $V_y[i]$ тезлик компоненталари $-V_m$ ва $+V_m$

оралигидаги кийматларга эришади. Зарралар координаталарини

экран четидан (яшик деворларидан) 10 бирлик узокда

жойлаштиришга сабаб - агар заррани дисплейда айлана курунишда

тасвирланса у экран ичида яккол куруниб туришидир.

3. Харакат тенгламасини интеграллаш

m массали i -номерли заррачага таъсир этувчи куч $F[i]$ булса

Нютон харакат тенгламаси куйидагича ёзилади

$$d^2x[i]/dt^2 = F_x [i]/m \quad (3)$$

$$d^2y[i]/dt^2 = F_y [i] /m \quad (4)$$

Иккинчи тартибли тенгламани иккита биринчи тартибли

тенгламага алмаштирамиз

$$dV_x[i]/dt = F_x[i]/m \quad (5)$$

$$dX[i]/dt = V_x[i] \quad (6)$$

$$dV_y[i]/dt = F_y[i]/m \quad (7)$$

$$dY[i]/dt = V_y[i] \quad (8)$$

Заррача массасини кийматини беришга хожат йук, (2) формулага асосан уни Z2 оркали ифодалаш мумкин. Бошқача килиб айтганда $m=1$ деб оламиз.

(5-8) тенгламаларни чекли орттирмалар курунишда ёзсак Эйлер алгоритмига келамиз, яъни

$$V_x[i]:=V_x[i] + F_x[i] * dt \quad (9)$$

$$V_y[i]:=V_y[i] + F_y[i]*dt \quad (10)$$

$$x[i]:=x[i]+V_x[i]*dt \quad (11)$$

$$y[i]:=y[i]+V_y[i]* dt \quad (12)$$

Демак dt вақтдан сунг заррачани тезликлари (9-10) ёрдамида, координаталари эса (11-12) ёрдамида хисобланади. Бунинг учун дастлаб i - заррачага таъсир этувчи куч (узидан бошка барча заррачалар томонидан таъсир этувчи кучлар йигиндиси) ни компоненталарини $F_x[i]$, $F_y[i]$ ни топиш керак.

Бу хисоблашлар куйидаги программада келтирилган

(D2)

For i:=1 to N do

begin

$F_x[i]:=0$; $F_y[i]:=0$;

For j:=1 to N do If $i \neq j$ then (i -заррачани узидан бошка барчаси томонидан)

```

begin r:=sqrt(sqr(x[i]-x[j])+sqr(y[i]-y[j]));
      Fx[i]:=Fx[i]+(x[i]-x[j])/(r*r*r);
      Fy[i]:=Fy[i]+(y[i]-y[j])/(r*r*r);
end;
Fx[i]:=g *Fx[i]; Fy[i]:=g*Fy[i];
end;

```

Тезликлари ва координаталари хисоблаш

(D3)

For i:=1 to N do

begin

Vx[i]:=Vx[i]+Fx[i]*dt; Vy[i]:=Vy[i]+Fy[i]*dt;

x[i]:= x[i]+Vx[i]*dt; y[i]:=y[i]+Vy[i]*dt;

end;

4. Заррачаларни яшик деворидан кайтиши

Куйидаги дасту агар заррача координатаси яшик деворидан утиб кетса (Экранны чегаралари 0, GetMaxX, 0, GetMaxY) уни харакат йуналиши тескарисига алмаштирилади яъни тезликлар ишораси узгартирилади

(D4)

For i:=1 to N do

begin

If x[i]< 0 then Vx[i]:= -Vx[i];

```

If y[i]<0          then Vy[i]:= -Vy[i];
If x[i]> GetMaxX  then Vx[i]:= -Vx[i];
If y[i]>GetMaxY  then Vy[i]:= -Vy[i];
end;

```

5. Экранда тасвирлаш. Мультипликация

Заррачани кучишини куриш учун аввал шу заррачани экранда тасвирлаш керак. Кучишдан сунг аввлги тасвирни учириб, янги координаталар буйича янги тасвир чизилади. Куйида программада

(D5)

```

t := t + dt;
If t > 50*dt then
For i:=1 to N do
begin SetColor(Black); Circle(Round( Xold[i]),Round( Yold[i] ), Rs );
      SetColor(i);Circle( Round( x[i]),Round( y[i] ),      Rs );
      Xold[i]:=x[i]; Yold[i]:=y[i];
      t:=0;
end;

```

заррачани R_s - радиусли айлана куринишда

```

SetColor(i);   Circle( Round( X[i]), Round( Y[i] ), Rs );

```

оператори ёрдамида чизилган. SetColor(i) - командаси i-заррани чизишда i - чи рангдан фойдаланади. Чизишдан олдин эски

координатали тасвир учирилган (тасвир кора рангда чизилган, экран асоси кора рангда бўлган хол учун)

```
SetColor(Black); Circle(Round( Xold[i]),Round( Yold[i] ), Rs );
```

Чизиб бўлгандан сунг эски координаталар сифатида янгиси кабул килинган

```
Xold[i]:=x[i]; Yold[i]:=y[i];
```

Заррани кучишини $t = 5..100 dt$ вақт оралигида (бу вақт оралиги компьютер процессорини ишлаш тезлигига боғлиқ) текшириб куриб мултипликацияни яхши намоён буладиган килиб танланади.

Саволлар

1. Даврий чегаравий шарт афзаллиги нимада? Унга программа тузинг.
2. Агар программада V_m киймати узгарса жараён кандай кечади?
Изохланг
3. Агар программада Z_2 киймати узгарса жараён кандай кечади?
Изохланг
4. Программада энергияни сакланишини текшириш мумкинми?
Кандай килиб ?

5. Системани совутиб буладими ? Кандай дастур ёрдамида ?
6. Агар ситема совутиб борилса нима кузатилиш керак ?
7. Ташки электр майдони таъсирини хисобга олиш мумкинми ?
кандай килиб ?

Адабиётлар

1. Р.Хокни, Дж. Иствуд. Численное моделирование мето дом частиц (Пер. с англ) Москва. "Мир". 1987. 640 с.
2. Д.В.Хеерман. Методы компьютерного эксперимента в теоретической физике. (Перевод с англ) Москва. "Наука". 1990. 176 с.
3. Х.Гулд, Я.Табочник. Компьютерное моделирование в физике. Часть 1. (Перевод с англ) Москва. "Мир". 1990 349 с.

Илова. Моделнинг Паскал тилидаги тула дастури

```

PROGRAMM Md_Charges1; { pbaymatov, Namangan, 2006 }
USES Crt,Dos,Graph;
CONST N=10; Rs=4; Z2=100; Vm=3; dt=0.02;
VAR Gd,Gm: Integer;
    x,y,Fx,Fy,Vx,Vy,Xold,Yold: Array[1..N] of Extended;
    r,t: Extended;
    i,j: byte;

```

```

BEGIN ClrScr;
Gd:=Detect; InitGraph(Gd,Gm,"); if GraphResult<>grOk then Halt(1);
Randomize;
For i:=1 to N do
begin x[i]:= 10 + Random( GetMaxX - 10);
      y[i]:= 10 + Random( GetMaxY - 10);
      Vx[i]:=Vm*( 2*0.01*Random(100) - 1 );
      Vy[i]:=Vm*( 2*0.01*Random(100) - 1 );
      Xold[i]:=x[i];
      Yold[i]:=y[i];
end;

t:=0;
Repeat
  For i:=1 to N do
  begin
    Fx[i]:=0; Fy[i]:=0;
    For j:=1 to N do If i<>j then
    begin r:=sqrt(sqr(x[i]-x[j])+sqr(y[i]-y[j]));
          Fx[i]:=Fx[i]+(x[i]-x[j])/(r*r*r);
          Fy[i]:=Fy[i]+(y[i]-y[j])/(r*r*r);
    end;
    Fx[i]:=Z2*Fx[i]; Fy[i]:=Z2*Fy[i];
  end;

  For i:=1 to N do
  begin
    Vx[i]:=Vx[i]+Fx[i]*dt; Vy[i]:=Vy[i]+Fy[i]*dt;

```

```

        x[i]:=x[i]+Vx[i]*dt; y[i]:=y[i]+Vy[i]*dt;
end;

For i:=1 to N do
begin
    If x[i]<0    then Vx[i]:=-Vx[i];
    If y[i]<0    then Vy[i]:=-Vy[i];
    If x[i]>GetMaxX then Vx[i]:=-Vx[i];
    If y[i]>GetMaxY then Vy[i]:=-Vy[i];
end;

t:=t+dt;
If t>50*dt then
For i:=1 to N do
begin SetColor(Black); Circle(Round( Xold[i]),Round( Yold[i] ), Rs );
      SetColor(i+5); Circle( Round( x[i]),Round( y[i] ),      Rs );
      Xold[i]:=x[i]; Yold[i]:=y[i];
      t:=0;
end;
Until KeyPressed;
CloseGraph;
END.

```


”